



16bits*16bits 无符号乘法运算

```
*****  
*****  
;程序名称: _16MUL16  
;程序功能: 16bits 乘以 16bits 无符号乘法运算子程序  
;入口参数: H_IER/L_IER 存放乘数  
;          ACCCH/ACCCL 存放被乘数  
;出口参数: ACCBH/ACCBL/ACCAH/ACCAL 存放 32 位运算结果(积)  
*****  
_16MUL16:  
    CLR        ACCAL  
    CLR        ACCAH  
    CLR        ACCBL  
    CLR        ACCBH  
    LDIA       D' 16'  
    LD         ACCNT, A  
_16MUL16_1:  
    RRCR       H_IER  
    RRCR       L_IER  
    SNZB       STATUS, C  
    JP         _16MUL16_2  
    LD         A, ACCCL  
    ADDR       ACCBL  
    SZB        STATUS, C  
    INCR       ACCBH  
    LD         A, ACCCH  
    ADDR       ACCBH
```



_16MUL16_2:

```

RRCR    ACCBH
RRCR    ACCBL
RRCR    ACCAH
RRCR    ACCAL
SZDECR  ACCNT
JP      _16MUL16_1
RET     00H

```

;应用实例

;使用前为子程序声明缓存变量(均为局部变量,可与其它局部变量复用)

```

H_IER    EQU    ?        ;乘数高八位
L_IER    EQU    ?        ;乘数低八位
ACCCH    EQU    ?        ;被乘数高八位
ACCCL    EQU    ?        ;被乘数低八位
ACCBH    EQU    ?        ;积 24-31 位
ACCBL    EQU    ?        ;积 16-23 位
ACCAH    EQU    ?        ;积 8-15 位
ACCAL    EQU    ?        ;积 0-7 位
ACCNT    EQU    ?        ;移位运算次数

```

;以下声明用户变量



USERRAM1 EQU ?

USERRAM2 EQU ?

;应用程序片段

;上接其它程序

SHOW_PROGRAM:

LDIA 055H

LD H_IER, A

LDIA 086H

LD L_IER, A ;赋值给乘数

LDIA 080H

LD ACCCH, A

LDIA 050H

LD ACCCL, A ;赋值给被被乘数

CALL _16MUL16 ;调用 16*16 子程序

LD A, ACCBH

LD USERRAM1, A

LD A, ACCBL

LD USERRAM2, A ;将积的高 16 位保存到用户寄存器

;下接其它程序
